*********************************************************************************
*** Appendix A5
*********************************************************************************

* Project: 	International threats and European defense
* Authors: 	Matthias Mader
* Date:		March 2023	

* Data: VW Project, Pane-European survey (W1)

*********************************************************************************

*** Setup

	clear
	set more off            						// Disable partitioned output
	set dp period									// Use comma instead of dot for decimals
	set linesize 80         						// Line size limit to make output more readable
	macro drop _all         						// clear all macros
	set scheme plotplainblind, permanent				// set graph scheme, e.g., cleanplots, plottig, plotplainblind, burd3, s1mono
	graph set window fontface "Times New Roman"		// set font in graphs | "Arial" | "Times New Roman"


*** Figure A5.1
	
	use DATA_clean, clear

	*Main effect of threat perceptions
	reg csdp_wouldlike c.threat_ind2 i.idconfig2 i.QCOUNTRY female age ib2.education iso comp mili feel_gov antielite feel_EU
	estimates store m1
	margins, at(threat_ind2=(0 (.025) 1))
	marginsplot,  ///
		name(g1, replace) ///
		recast(scatter) ///
		plotopts(mcolor(none))  ///
		ciopts(recast(rbar) barwidth(0.01)) ///
		xtitle("Threat perception", size(large)) xscale(range(0 1)) xlabel(0 (0.2) 1, labsize(large)) ///	
		ytitle("Support for integration", size(large)) yscale(range(0 1)) ylabel(0 (0.2) 1, labsize(large)) ///	
		title("  ") ///
		legend(off)

	*Support for European defence and security integration, by threat perception and EU attitude
	reg csdp_wouldlike c.threat_ind2##i.idconfig2 i.QCOUNTRY female age ib2.education iso comp mili feel_gov antielite feel_EU
	estimates store m2
	margins, at(threat_ind2=(0 (.025) 1) idconfig2=(1 2))
	marginsplot,  ///
		name(g2, replace) ///
		recast(scatter) ///
		plot1opts(mcolor(none)) plot2opts(mcolor(none)) ///
		text(0.85 -0.01 "Inclusive identity", color(gs9) place(right) size(large)) ///
		text(0.40 -0.01  "Exclusive national identity", color(gs0) place(right) size(large)) ///
		ciopts(recast(rbar) barwidth(0.01)) ///
		xtitle("Threat perception", size(large)) xscale(range(0 1)) xlabel(0 (0.2) 1, labsize(large)) ///	
		ytitle("Support for integration", size(large)) yscale(range(0 1)) ylabel(0 (0.2) 1, labsize(large)) ///	
		title("  ") ///
		legend(off)


	*** Text: Decreased attitude gap between EU opponents and supporters
	margins, at(threat_ind2=(0 1) idconfig2=(1 2))

	*** Text: Effect sizes for EU opponents and supporters
	margins, dydx(threat_ind2) at(idconfig2=(1 2))


	*Support for European defence and security integration among the politically unsophisticated, by threat perception and EU attitude
	reg csdp_wouldlike c.threat_ind2##i.idconfig2##c.quizr i.QCOUNTRY female age ib2.education iso comp mili feel_gov antielite feel_EU
	estimates store m3
	margins, at(threat_ind2=(0 (.025) 1) idconfig2=(1 2) quizr=(0))
	marginsplot,  ///
		name(g3, replace) ///
		recast(scatter) ///
		plot1opts(mcolor(none)) plot2opts(mcolor(none)) ///
		text(0.80 -0.01 "Inclusive identity", color(gs9) place(right) size(large)) ///
		text(0.50 -0.01  "Exclusive national identity", color(gs0) place(right) size(large)) ///
		ciopts(recast(rbar) barwidth(0.01)) ///
		xtitle("Threat perception", size(large)) xscale(range(0 1)) xlabel(0 (0.2) 1, labsize(large)) ///	
		ytitle("Support for integration", size(large)) yscale(range(0 1)) ylabel(0 (0.2) 1, labsize(large)) ///	
		title("LOW SOPHISTICATION", size(large))  ///
		legend(off)

	*Text: Decreased attitude gap between Europhiles and Europhobes
	margins, dydx(threat_ind2) at(idconfig2=(1 2) quiz=(0 4))


	*Figure: Support for European defence and security integration among politically sophisticated, by threat perception and EU attitude
	reg csdp_wouldlike c.threat_ind2##i.idconfig2##c.quizr i.QCOUNTRY female age ib2.education iso comp mili feel_gov antielite feel_EU
	margins, at(threat_ind2=(0 (.025) 1) idconfig2=(1 2) quizr=(1))
	marginsplot,  ///
		name(g4, replace) ///
		recast(scatter) ///
		plot1opts(mcolor(none)) plot2opts(mcolor(none)) ///
		text(0.85 -0.01 "Inclusive identity", color(gs9) place(right) size(large)) ///
		text(0.43 -0.01  "Exclusive national identity", color(gs0) place(right) size(large)) ///
		ciopts(recast(rbar) barwidth(0.01)) ///
		xtitle("Threat perception", size(large)) xscale(range(0 1)) xlabel(0 (0.2) 1, labsize(large)) ///	
		ytitle("Support for integration", size(large)) yscale(range(0 1)) ylabel(0 (0.2) 1, labsize(large)) ///	
		title("HIGH SOPHISTICATION", size(large))  ///
		legend(off)

